

1 Introduction

This document introduces the ARM* Generic Interrupt Controller (GIC), which is included as part of the ARM Cortex-A9* MPCORE* processor in the Intel® Cyclone® V SoC family. We do not discuss some of the advanced features of the GIC in this document; complete information is available in the publication entitled *ARM Generic Interrupt Controller Architectural Specification*, which is available from ARM Holdings.

Contents:

- Purpose of the GIC
- ARM Exception Processing Architecture
- GIC Architecture
- GIC Programmer's Interface
- Examples of ARM Software Code for the GIC

2 ARM* Generic Interrupt Controller

As illustrated in Figure 1, the ARM *generic interrupt controller* (GIC) is a part of the ARM A9 MPCORE processor. The GIC is connected to the IRQ interrupt signals of all I/O peripheral devices that are capable of generating interrupts. Most of these devices are normally external to the A9 MPCORE, and some are internal peripherals (such as timers). The GIC included with the A9 MPCORE processor in the Intel Cyclone V SoC family handles up to 255 sources of interrupts. When a peripheral device sends its IRQ signal to the GIC, then the GIC can forward a corresponding IRQ signal to one or both of the A9 cores. Software code that is running on the A9 core can then query the GIC to determine which peripheral device caused the interrupt, and take appropriate action. The procedure for working with interrupts for the ARM Cortex-A9 and the GIC are described in the following sections.




Figure 1. The ARM A9 MPCORE processor.

3 Interrupts in the ARM Cortex-A9*

An introduction to ARM processors can be found in the tutorial *Introduction to the ARM Processor Using Intel/ARM Toolchain*, which is available on Intel's FPGA University Program website. As described in that tutorial, the ARM Cortex-A9 has several main modes of operation, listed below:

- *User* mode – is the basic mode in which application programs run. This is an unprivileged mode, which has restricted access to system resources.
- *System* mode – provides full access to system resources. It can be entered only from one of the exception modes listed below.
- *Supervisor* mode – is entered when the processor executes a *supervisor call* instruction, SVC. It is also entered on reset or power-up.

- *Abort* mode – is entered if the processor attempts to access a non-legitimate memory location. This can happen, for example, when performing a word access for an address that is not word-aligned.
- *Undefined* mode – is entered if the processor attempts to execute an unimplemented instruction.
- *IRQ* mode – is entered in response to an interrupt request.
- *FIQ* mode – is entered in response to a *fast interrupt* request. We do not discuss fast interrupts in this document; they are used in some Cortex-A9 systems to provide faster service for more urgent requests. This document focuses only on IRQ interrupts.

When the processor is first powered on, or reset, it is in the *Supervisor* mode. This mode is *privileged*, which means that it allows the use of all processor instructions and operations. From supervisor mode it is possible to change into *User* mode, which is the only non-privileged mode. In User mode certain types of processor operations and instructions are prohibited. In practice, the Supervisor mode is normally used when the processor is executing software such as an operating system, whereas other software code may run in the User mode, thereby providing a level of protection for critical resources.

The operating mode of the processor is indicated in the current processor status register CPSR, as depicted in Figure 2. The mode bits are defined in Table 1.




Figure 2. The current processor status register (CPSR).

TABLE 1. Mode Bits

| CPSR ₄₋₀ | Operating Mode |
|---------------------|----------------|
| 10000 | User |
| 10001 | FIQ |
| 10010 | IRQ |
| 10011 | Supervisor |
| 10111 | Abort |
| 11011 | Undefined |
| 11111 | System |

To manipulate the contents of the CPSR, the processor must be in one of the privileged modes. Figure 3 shows the general-purpose registers in a Cortex-A9 processor, and illustrates how the registers are related to the processor mode. In User mode, there are 16 registers, *R0 – R15*, plus the CPSR. These registers are also available in the System mode, which is not shown in the figure. As indicated in Figure 3, *R0 – R12*, as well as the program counter *R15*, are common to all modes except FIQ. But the stack pointer register *R13* and the link register *R14* are not common—*banked* versions of these registers exist for each mode. Thus, the Supervisor mode has a stack pointer and link register that are used only when the processor is in this mode. Similarly, the other modes, such as IRQ mode, have their own stack pointers and link registers. The CPSR register is common for all modes, but when the processor is switched from one mode into another, the current content of the CPSR is copied into the new mode’s saved processor status register (SPSR). Note that the FIQ mode, which we do not discuss in this document, has the additional banked registers *R8 – R12*, as shown in the figure.

| | User | Supervisor | Abort | Undefined | IRQ | FIQ |
|----|------|------------|----------|-----------|----------|----------|
| | R0 | R0 | R0 | R0 | R0 | R0 |
| | R1 | R1 | R1 | R1 | R1 | R1 |
| | R2 | R2 | R2 | R2 | R2 | R2 |
| | R3 | R3 | R3 | R3 | R3 | R3 |
| | R4 | R4 | R4 | R4 | R4 | R4 |
| | R5 | R5 | R5 | R5 | R5 | R5 |
| | R6 | R6 | R6 | R6 | R6 | R6 |
| | R7 | R7 | R7 | R7 | R7 | R7 |
| | R8 | R8 | R8 | R8 | R8 | R8_fiq |
| | R9 | R9 | R9 | R9 | R9 | R9_fiq |
| | R10 | R10 | R10 | R10 | R10 | R10_fiq |
| | R11 | R11 | R11 | R11 | R11 | R11_fiq |
| | R12 | R12 | R12 | R12 | R12 | R12_fiq |
| SP | R13 | R13_svc | R13_abt | R13_und | R13_irq | R13_fiq |
| LR | R14 | R14_svc | R14_abt | R14_und | R14_irq | R14_fiq |
| PC | R15 | R15 | R15 | R15 | R15 | R15 |
| | CPSR | CPSR | CPSR | CPSR | CPSR | CPSR |
| | | SPSR_svc | SPSR_abt | SPSR_und | SPSR_irq | SPSR_fiq |

Figure 3. Banked registers in ARM processors.

3.1 IRQ Mode

A Cortex-A9 processor enters IRQ mode in response to receiving an IRQ signal from the GIC. Before such interrupts can be used, software code has to perform a number of steps:

1. Ensure that IRQ interrupts are disabled in the A9 processor, by setting the IRQ disable bit in the CPSR to 1.
2. Configure the GIC. Interrupts for each I/O peripheral device that is connected to the GIC are identified by a unique *interrupt ID*.

3. Configure each I/O peripheral device so that it can send IRQ interrupt requests to the GIC.
4. Enable IRQ interrupts in the A9 processor, by setting the IRQ disable bit in the CPSR to 0.

Examples of software code that perform these steps are given in Sections 5 and 6. Complete examples of interrupt-driven code are included in the appendices.

4 Programmer’s Interface to the GIC

The GIC includes a number of memory-mapped registers that provide an *application programmer’s interface* (API). As illustrated in Figure 4, the GIC architecture is divided into two main parts, called the *CPU Interface* and the *Distributor*. The CPU Interface is responsible for sending IRQ requests received by the Distributor to one or both of the A9 processors in the MPCORE. The Distributor receives IRQ interrupt signals from I/O peripherals.




Figure 4. The GIC Architecture.

4.1 GIC CPU Interface

The CPU Interface in the GIC is used to send IRQ signals to the A9 cores. There is one CPU Interface for each A9 core in the MPCORE. API registers in each CPU Interface are depicted in Figure 5. To make the example more concrete, we have assigned addresses to these registers, as shown. These addresses correspond to those used in the document *DE1-SoC Computer System with ARM Cortex-A9*, which is available from Intel’s FPGA University Program. The DE1-SoC Computer System is an ARM Cortex-A9 embedded system that can be implemented on Intel’s DE1-SoC development and education board.

The *CPU Interface Control Register* (ICCICR) is used to enable forwarding of interrupts from the CPU Interface to the corresponding A9 core. Setting bit *E* = 1 in this register enables the sending of interrupts to the A9 core, and setting *E* = 0 disables these interrupts.

The *Interrupt Priority Mask Register (ICCPMR)* is used to set a threshold for the priority-level of interrupts that will be forwarded by a CPU Interface to an A9 core. Only interrupts that have a priority level greater than the *Priority* field in ICCPMR will be sent to an A9 processor by its CPU Interface. Lower priority values represent higher priority, meaning that level 0 is the highest priority and level 255 is the lowest. Setting the *Priority* field in ICCPMR to the value 0 will prevent any interrupts from being generated by the CPU Interface. The procedure for setting the priority level of individual interrupts (based on their Interrupt ID) is described in Section 4.2.

The *Interrupt Acknowledge Register (ICCIAR)* contains the Interrupt ID of the I/O peripheral that has caused an interrupt. When an A9 processor receives an IRQ signal from the GIC, software code (i.e., the *interrupt handler*) running on the processor must read the ICCIAR to determine which I/O peripheral has caused the interrupt.

After the A9 processor has completed the handling of an IRQ interrupt generated by the GIC, the processor must then clear this interrupt from the CPU Interface. This action is accomplished by writing the appropriate Interrupt ID into the *Interrupt ID* field in the *End of Interrupt Register (ICCEOIR)*, depicted in Figure 5. After writing into the ICCEOIR, the interrupt handler software can then return control to the previously-interrupted main program.

| Address | 31 | ... | 10 | 9 | 8 | 7 | ... | 1 | 0 | Register name | |
|-----------|--------|-----|----|---|---|---|--------------|---|---|---------------|--------|
| 0xFFFE100 | Unused | | | | | | | | | E | ICCIAR |
| 0xFFFE104 | Unused | | | | | | Priority | | | ICCPMR | |
| 0xFFFE10C | Unused | | | | | | Interrupt ID | | | ICCIAR | |
| 0xFFFE110 | Unused | | | | | | Interrupt ID | | | ICCEOIR | |

Figure 5. CPU Interface registers.

4.2 GIC Distributor

The Distributor in the GIC can handle 255 sources of interrupts. As indicated in Figure 4, Interrupt IDs in the range from 32 – 255 correspond to *shared peripheral interrupts (SPIs)*. These interrupts are connected to the IRQ signals of up to 224 I/O peripherals, and these sources of interrupts are common to (shared by) both CPU Interfaces. The Distributor also handles *private peripherals interrupts (PPIs)* for each of the A9 processors, with these interrupts using IDs in the range from 0 – 31. The *software generated interrupts (SGIs)* are a special type of private interrupt that are generated by writing to a specific register in the GIC; Interrupt IDs from 0 – 15 are used for SGIs. We do not discuss SGIs further in this document.

API registers in the Distributor are depicted in Figure 6. As described in the previous section, addresses are shown for each register and these addresses correspond to those used in the DE1-SoC Computer. The Distributor Control Register (ICDDCR) is used to enable the Distributor. Setting $E = 0$ in this register disables the Distributor, while setting $E = 1$ enables it.

The *Interrupt Set Enable Registers (ICDISERn)* are used to enable the forwarding of each supported interrupt from the Distributor to the CPU Interface. The n postfix in the name ICDISERn means that multiple registers exist. Refer-

| Base Address | 31 | ... | 24 | 23 | ... | 16 | 15 | ... | 8 | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 | Register name |
|--------------|--------------------|-----|-----|-----|--------------------|-----|----|-----|--------------------|----|----|----|--------------------|----|----|----|---|---------------|
| 0xFFFD000 | Unused | | | | | | | | | | | | | | | | E | ICDDCR |
| 0xFFFD100 | Set-enable bits | | | | | | | | | | | | | | | | | ICDISERn |
| ... | Set-enable bits | | | | | | | | | | | | | | | | | ... |
| 0xFFFD180 | Clear-enable bits | | | | | | | | | | | | | | | | | ICDICERn |
| ... | Clear-enable bits | | | | | | | | | | | | | | | | | ... |
| 0xFFFD400 | Priority, offset 3 | | | | Priority, offset 2 | | | | Priority, offset 1 | | | | Priority, offset 0 | | | | | ICDIPRn |
| ... | Priority, offset 3 | | | | Priority, offset 2 | | | | Priority, offset 1 | | | | Priority, offset 0 | | | | | ... |
| 0xFFFD800 | CPUs, offset 3 | | | | CPUs, offset 2 | | | | CPUs, offset 1 | | | | CPUs, offset 0 | | | | | ICDIPTRn |
| ... | CPUs, offset 3 | | | | CPUs, offset 2 | | | | CPUs, offset 1 | | | | CPUs, offset 0 | | | | | ... |
| 0xFFFDC00 | F15 | F14 | F13 | F12 | F11 | F10 | F9 | F8 | F7 | F6 | F5 | F4 | F3 | F2 | F1 | F0 | | ICDICFRn |
| ... | F15 | F14 | F13 | F12 | F11 | F10 | F9 | F8 | F7 | F6 | F5 | F4 | F3 | F2 | F1 | F0 | | ... |

Figure 6. Distributor registers.

ring to Figure 6, the set-enable bits for the first 32 Interrupt IDs are provided in the register at address 0xFFFD100, the next 32 are provided in the register at the following word address, which is 0xFFFD104, and so on. Given a specific Interrupt ID, N , the address of the register that contains its set-enable bit is given by the integer calculation $address = 0xFFFD100 + (N \div 32) \times 4$, and the index of the bit inside this register is given by $index = N \bmod 32$. Writing the value 1 into a set-enable bit enables the forwarding of the corresponding IRQ to the CPU Interface.

In the same way that each supported interrupt can be enabled by using ICDISERn, each interrupt can be disabled by using the *Interrupt Clear Enable Registers* (ICDICERn). The method for calculating the address and index for ICDICERn is the same as that for ICDISERn, except that the base address is 0xFFFD180, as shown in Figure 6. Writing a 1 into a clear-enable bit disables the forwarding of the corresponding interrupt to the CPU Interface.

The *Interrupt Priority Registers* (ICDIPRn) are used to associate a priority level with each individual interrupt. On reset, these registers are set to 0x00000000, which represents the highest priority. In Figure 6 the base address of ICDIPRn is 0xFFFD400. Each Interrupt ID's priority field is one byte in size, which means that the register at the base address holds the priority levels for Interrupt IDs from 0 to 3. The priority levels for the next four Interrupt IDs use the register at address 0xFFFD404, and so on. Given a specific Interrupt ID, N , the address of the register that contains its priority field is given by the integer calculation $address = 0xFFFD400 + (N \div 4) \times 4$, and the index of the byte inside this register is given by $index = N \bmod 4$. Setting the priority field for an Interrupt ID to a larger number results in lower priority for the corresponding interrupt.

The *Interrupt Processor Targets Registers* (ICDIPTRn) are used to specify the CPU interfaces to which each interrupt should be forwarded. As indicated in Figure 6, the *CPUs* field for each Interrupt ID is one byte in size. This size is used because some versions of the ARM A9 MPCORE have up to eight A9 cores. A target CPU is selected by setting its corresponding bit field to 1. Thus, setting the byte at address 0xFFFFED800 to the value 0x01 would target Interrupt ID 0 to CPU 0, setting this same byte to 0x02 would target CPU 1, and setting the byte to the value 0x03 would target both CPU 0 and CPU 1. The scheme for calculating the address of the ICDIPTRn register for a specific Interrupt ID, and also its byte index, is the same as the one shown above for ICDIPRn.

The *Interrupt Configuration Registers* (ICDICFRn) are used to specify whether each supported interrupt should be handled as level- or edge-sensitive by the GIC. As indicated in Figure 6, there is a two-bit field associated with each Interrupt ID. The least-significant bit in this field is not used. Setting the most-significant bit of this field to 1 makes the corresponding interrupt signal edge-sensitive, and setting this field to 0 makes it level-sensitive. When a level-sensitive IRQ signal is asserted by an I/O peripheral it is possible to de-assert this signal if the interrupt has not yet been forwarded from the Distributor to a CPU Interface. However, an edge-triggered IRQ signal cannot be de-asserted once it has been sampled in the Distributor. Referring to Figure 6, the first 16 Interrupt IDs use the ICDICFRn register at address 0xFFFFEDC00, the next 16 at address 0xFFFFEDC04, and so on. Given a specific Interrupt ID, N , the address of the ICDICFRn register is given by the integer calculation $address = 0xFFFFEDC00 + (N \div 16) \times 4$, and the index of the bit inside this register is given by $index = (N \bmod 16) + 1$.

5 Example of Assembly Language Code

Figure 7 provides an example of an assembly language subroutine that configures the GIC. This code configures Interrupt ID 73, as an example, which corresponds to a parallel port connected to pushbutton KEYS in the DE1-SoC Computer. The code configures only some of the registers in the GIC and uses acceptable default values for other registers. A complete example of code that uses this subroutine is provided in the Appendix A.

```

/*
 * Configure the Generic Interrupt Controller (GIC)
 */
    .global CONFIG_GIC
CONFIG_GIC:
    PUSH    {LR}
    /* To configure the FPGA KEYS interrupt (ID 73):
     * 1. set the target to cpu0 in the ICDIPTRn register
     * 2. enable the interrupt in the ICDISERn register */
    /* CONFIG_INTERRUPT (int_ID (R0), CPU_target (R1)); */
    MOV     R0, #73                // KEY port (Interrupt ID = 73)
    MOV     R1, #1                // this field is a bit-mask; bit 0 targets cpu0
    BL     CONFIG_INTERRUPT

    /* configure the GIC CPU Interface */
    LDR     R0, =0xFFFE100        // base address of CPU Interface
    /* Set Interrupt Priority Mask Register (ICCPMR) */
    LDR     R1, =0xFFFF          // enable interrupts of all priorities levels
    STR     R1, [R0, #0x04]
    /* Set the enable bit in the CPU Interface Control Register (ICCICR).
     * This allows interrupts to be forwarded to the CPU(s) */
    MOV     R1, #1
    STR     R1, [R0]

    /* Set the enable bit in the Distributor Control Register (ICDDCR).
     * This enables forwarding of interrupts to the CPU Interface(s) */
    LDR     R0, =0xFFFE000
    STR     R1, [R0]

    POP     {PC}

```

Figure 7. An example of assembly language code that configures the GIC (Part a).

```

/*
 * Configure registers in the GIC for an individual Interrupt ID
 * We configure only the Interrupt Set Enable Registers (ICDISERn) and
 * Interrupt Processor Target Registers (ICDIPTRn). The default (reset)
 * values are used for other registers in the GIC
 * Arguments: R0 = Interrupt ID (N), R1 = CPU target
 */
CONFIG_INTERRUPT:
    PUSH    {R4-R5, LR}

    /* Configure Interrupt Set-Enable Registers (ICDISERn).
     * reg_offset = (integer_div(N / 32) * 4
     * value = 1 << (N mod 32) */
    LSR     R4, R0, #3           // calculate reg_offset
    BIC     R4, R4, #3           // R4 = reg_offset
    LDR     R2, =0xFFFED100
    ADD     R4, R2, R4           // R4 = address of ICDISER

    AND     R2, R0, #0x1F       // N mod 32
    MOV     R5, #1               // enable
    LSL     R2, R5, R2          // R2 = value

    /* Using the register address in R4 and the value in R2 set the
     * correct bit in the GIC register */
    LDR     R3, [R4]            // read current register value
    ORR     R3, R3, R2          // set the enable bit
    STR     R3, [R4]            // store the new register value

    /* Configure Interrupt Processor Targets Register (ICDIPTRn)
     * reg_offset = integer_div(N / 4) * 4
     * index = N mod 4 */
    BIC     R4, R0, #3           // R4 = reg_offset
    LDR     R2, =0xFFFED800
    ADD     R4, R2, R4           // R4 = word address of ICDIPTR
    AND     R2, R0, #0x3        // N mod 4
    ADD     R4, R2, R4           // R4 = byte address in ICDIPTR

    /* Using register address in R4 and the value in R2 write to
     * (only) the appropriate byte */
    STRB    R1, [R4]

    POP     {R4-R5, PC}

```

Figure 7. An example of assembly language code that configures the GIC (Part b).

6 Example of C Code

Figure 8 provides an example of a subroutine written in C code that configures the GIC. This code performs the same operations as the assembly language code shown in Figure 7. A complete program that uses this subroutine is provided in the Appendix B.

```

/*
 * Configure the Generic Interrupt Controller (GIC)
 */
void config_GIC(void) {
    config_interrupt (73, 1); // configure the FPGA KEYS interrupt (73)

    // Set Interrupt Priority Mask Register (ICCPMR). Enable interrupts of all
    // priorities
    *((int *) 0xFFFFEC104) = 0xFFFF;

    // Set CPU Interface Control Register (ICCICR). Enable signaling of
    // interrupts
    *((int *) 0xFFFFEC100) = 1;

    // Configure the Distributor Control Register (ICDDCR) to send pending
    // interrupts to CPUs
    *((int *) 0xFFFFED000) = 1;
}

/*
 * Configure Set Enable Registers (ICDISERn) and Interrupt Processor Target
 * Registers (ICDIPTRN). The default (reset) values are used for other registers
 * in the GIC.
 */
void config_interrupt(int N, int CPU_target) {
    int reg_offset, index, value, address;

    /* Configure the Interrupt Set-Enable Registers (ICDISERn).
     * reg_offset = (integer_div(N / 32) * 4
     * value = 1 << (N mod 32) */
    reg_offset = (N >> 3) & 0xFFFFFFF4;
    index      = N & 0x1F;
    value      = 0x1 << index;
    address    = 0xFFFFED100 + reg_offset;
    /* Now that we know the register address and value, set the appropriate bit */
    *((int *)address) |= value;
}

```

Figure 8. An example of C language code that configures the GIC (Part a).

```
/* Configure the Interrupt Processor Targets Register (ICDIPTRn)
 * reg_offset = integer_div(N / 4) * 4
 * index = N mod 4 */
reg_offset = (N & 0xFFFFFFFFFC);
index      = N & 0x3;
address    = 0xFFFFED800 + reg_offset + index;
/* Now that we know the register address and value, write to (only) the
 * appropriate byte */
*(char *)address = (char)CPU_target;
}
```

Figure 8. An example of C code that configures the GIC (Part b).

Appendix A: Example Assembly Language Program

```

/* *****
 * This program demonstrates use of interrupts with assembly language code.
 * The program responds to interrupts from the pushbutton KEY port in the FPGA.
 *
 * The interrupt service routine for the pushbutton KEYS indicates which KEY has
 * been pressed on the HEX0 display.
 *****/

        .section .vectors, "ax"
        B        _start                // reset vector
        B        SERVICE_UND           // undefined instruction vector
        B        SERVICE_SVC           // software interrupt vector
        B        SERVICE_ABT_INST      // aborted prefetch vector
        B        SERVICE_ABT_DATA      // aborted data vector
        .word    0                     // unused vector
        B        SERVICE_IRQ           // IRQ interrupt vector
        B        SERVICE_FIQ          // FIQ interrupt vector

        .text
        .global  _start

_start:
        /* Set up stack pointers for IRQ and SVC processor modes */
        MOV     R1, #0b11010010        // interrupts masked, MODE = IRQ
        MSR     CPSR_c, R1             // change to IRQ mode
        LDR     SP, =0xFFFFFFFF - 3    // set IRQ stack to A9 onchip memory
        /* Change to SVC (supervisor) mode with interrupts disabled */
        MOV     R1, #0b11010011        // interrupts masked, MODE = SVC
        MSR     CPSR, R1               // change to supervisor mode
        LDR     SP, =0x3FFFFFFF - 3    // set SVC stack to top of DDR3 memory

        BL      CONFIG_GIC            // configure the ARM GIC
        /* Write to the pushbutton KEY interrupt mask register */
        LDR     R0, =0xFF200050        // pushbutton KEY base address
        MOV     R1, #0xF               // set interrupt mask bits
        STR     R1, [R0, #0x8]         // interrupt mask register (base + 8)
        /* Enable IRQ interrupts in the processor */
        MOV     R0, #0b01010011        // IRQ unmasked, MODE = SVC
        MSR     CPSR_c, R0

IDLE:
        B        IDLE                  // main program simply idles

        /* Define the exception service routines */

        /*--- Undefined instructions -----*/
SERVICE_UND:
        B        SERVICE_UND

        /*--- Software interrupts -----*/
SERVICE_SVC:

```

```

        B            SERVICE_SVC

/*--- Aborted data reads -----*/
SERVICE_ABT_DATA:
        B            SERVICE_ABT_DATA

/*--- Aborted instruction fetch -----*/
SERVICE_ABT_INST:
        B            SERVICE_ABT_INST

/*--- IRQ -----*/
SERVICE_IRQ:
        PUSH        {R0-R7, LR}

        /* Read the ICCIAR from the CPU Interface */
        LDR         R4, =0xFFFFEC100
        LDR         R5, [R4, #0x0C]      // read from ICCIAR

FPGA_IRQ1_HANDLER:
        CMP         R5, #73
UNEXPECTED:
        BNE         UNEXPECTED          // if not recognized, stop here

        BL         KEY_ISR
EXIT_IRQ:
        /* Write to the End of Interrupt Register (ICCEOIR) */
        STR         R5, [R4, #0x10]     // write to ICCEOIR

        POP        {R0-R7, LR}
        SUBS        PC, LR, #4

/*--- FIQ -----*/
SERVICE_FIQ:
        B            SERVICE_FIQ

        .end

/*
 * Configure the Generic Interrupt Controller (GIC)
 */
        .global CONFIG_GIC
CONFIG_GIC:
        PUSH        {LR}
        /* To configure the FPGA KEYS interrupt (ID 73):
         * 1. set the target to cpu0 in the ICDIPTRn register
         * 2. enable the interrupt in the ICDISERn register */
        /* CONFIG_INTERRUPT (int_ID (R0), CPU_target (R1)); */
        MOV         R0, #73             // KEY port (Interrupt ID = 73)
        MOV         R1, #1             // this field is a bit-mask; bit 0 targets cpu0
        BL         CONFIG_INTERRUPT

        /* configure the GIC CPU Interface */

```

```

LDR      R0, =0xFFFFEC100      // base address of CPU Interface
/* Set Interrupt Priority Mask Register (ICCPMR) */
LDR      R1, =0xFFFF           // enable interrupts of all priorities levels
STR      R1, [R0, #0x04]
/* Set the enable bit in the CPU Interface Control Register (ICCICR).
 * This allows interrupts to be forwarded to the CPU(s) */
MOV      R1, #1
STR      R1, [R0]

/* Set the enable bit in the Distributor Control Register (ICDDCR).
 * This enables forwarding of interrupts to the CPU Interface(s) */
LDR      R0, =0xFFFFED000
STR      R1, [R0]

POP      {PC}

/*
 * Configure registers in the GIC for an individual Interrupt ID
 * We configure only the Interrupt Set Enable Registers (ICDISERn) and
 * Interrupt Processor Target Registers (ICDIPTRn). The default (reset)
 * values are used for other registers in the GIC
 * Arguments: R0 = Interrupt ID (N), R1 = CPU target
 */
CONFIG_INTERRUPT:
    PUSH    {R4-R5, LR}

    /* Configure Interrupt Set-Enable Registers (ICDISERn).
     * reg_offset = (integer_div(N / 32) * 4
     * value = 1 << (N mod 32) */
    LSR     R4, R0, #3           // calculate reg_offset
    BIC     R4, R4, #3           // R4 = reg_offset
    LDR     R2, =0xFFFFED100
    ADD     R4, R2, R4           // R4 = address of ICDISER

    AND     R2, R0, #0x1F       // N mod 32
    MOV     R5, #1               // enable
    LSL     R2, R5, R2          // R2 = value

    /* Using the register address in R4 and the value in R2 set the
     * correct bit in the GIC register */
    LDR     R3, [R4]            // read current register value
    ORR     R3, R3, R2          // set the enable bit
    STR     R3, [R4]            // store the new register value

    /* Configure Interrupt Processor Targets Register (ICDIPTRn)
     * reg_offset = integer_div(N / 4) * 4
     * index = N mod 4 */
    BIC     R4, R0, #3           // R4 = reg_offset
    LDR     R2, =0xFFFFED800
    ADD     R4, R2, R4           // R4 = word address of ICDIPTR
    AND     R2, R0, #0x3        // N mod 4

```

```

ADD      R4, R2, R4           // R4 = byte address in ICDIPTR

/* Using register address in R4 and the value in R2 write to
 * (only) the appropriate byte */
STRB     R1, [R4]

POP      {R4-R5, PC}

/*****
 * Pushbutton - Interrupt Service Routine
 *
 * This routine checks which KEY has been pressed. It writes to HEX0
 *****/

.global  KEY_ISR
KEY_ISR:
LDR      R0, =0xFF200050 // base address of pushbutton KEY port
LDR      R1, [R0, #0xC]  // read edge capture register
MOV      R2, #0xF
STR      R2, [R0, #0xC] // clear the interrupt

LDR      R0, =0xFF200020 // based address of HEX display
CHECK_KEY0:
MOV      R3, #0x1
ANDS     R3, R3, R1      // check for KEY0
BEQ      CHECK_KEY1
MOV      R2, #0b00111111
STR      R2, [R0]       // display "0"
B        END_KEY_ISR

CHECK_KEY1:
MOV      R3, #0x2
ANDS     R3, R3, R1      // check for KEY1
BEQ      CHECK_KEY2
MOV      R2, #0b00000110
STR      R2, [R0]       // display "1"
B        END_KEY_ISR

CHECK_KEY2:
MOV      R3, #0x4
ANDS     R3, R3, R1      // check for KEY2
BEQ      IS_KEY3
MOV      R2, #0b01011011
STR      R2, [R0]       // display "2"
B        END_KEY_ISR

IS_KEY3:
MOV      R2, #0b01001111
STR      R2, [R0]       // display "3"
END_KEY_ISR:
BX      LR

.end

```


Appendix B: Example C Program

```

void disable_A9_interrupts(void);
void set_A9_IRQ_stack(void);
void config_GIC(void);
void config_KEYS(void);
void enable_A9_interrupts(void);

/* *****
 * This program demonstrates use of interrupts with C code. The program
 * responds
 * to interrupts from the pushbutton KEY port in the FPGA.
 *
 * The interrupt service routine for the KEYS indicates which KEY has been
 * pressed
 * on the LED display.
 ******/
int main(void) {
    disable_A9_interrupts(); // disable interrupts in the A9 processor
    set_A9_IRQ_stack();     // initialize the stack pointer for IRQ mode
    config_GIC();           // configure the general interrupt controller
    config_KEYS();          // configure pushbutton KEYS to generate interrupts

    enable_A9_interrupts(); // enable interrupts in the A9 processor

    while (1) // wait for an interrupt
        ;
}

/* setup the KEY interrupts in the FPGA */
void config_KEYS() {
    volatile int * KEY_ptr = (int *) 0xFF200050; // pushbutton KEY base address

    *(KEY_ptr + 2) = 0xF; // enable interrupts for the two KEYS
}

/* This file:
 * 1. defines exception vectors for the A9 processor
 * 2. provides code that sets the IRQ mode stack, and that dis/enables
 * interrupts
 * 3. provides code that initializes the generic interrupt controller
 */
void pushbutton_ISR(void);
void config_interrupt(int, int);

// Define the IRQ exception handler
void __attribute__((interrupt)) __cs3_isr_irq(void) {
    // Read the ICCIAR from the CPU Interface in the GIC
    int interrupt_ID = *((int *)0xFFFFEC10C);

    if (interrupt_ID == 73) // check if interrupt is from the KEYS
        pushbutton_ISR();
}

```

```

else
    while (1);           // if unexpected, then stay here

    // Write to the End of Interrupt Register (ICCEOIR)
    *((int *)0xFFFECC110) = interrupt_ID;
}

// Define the remaining exception handlers
void __attribute__((interrupt)) __cs3_reset(void) {
    while (1);
}

void __attribute__((interrupt)) __cs3_isr_undef(void) {
    while (1);
}

void __attribute__((interrupt)) __cs3_isr_swi(void) {
    while (1);
}

void __attribute__((interrupt)) __cs3_isr_pabort(void) {
    while (1);
}

void __attribute__((interrupt)) __cs3_isr_dabort(void) {
    while (1);
}

void __attribute__((interrupt)) __cs3_isr_fiq(void) {
    while (1);
}

/*
 * Turn off interrupts in the ARM processor
 */
void disable_A9_interrupts(void) {
    int status = 0b11010011;
    asm("msr cpsr, %[ps]" : : [ps] "r"(status));
}

/*
 * Initialize the banked stack pointer register for IRQ mode
 */
void set_A9_IRQ_stack(void) {
    int stack, mode;
    stack = 0xFFFFFFFF - 7; // top of A9 onchip memory, aligned to 8 bytes
    /* change processor to IRQ mode with interrupts disabled */
    mode = 0b11010010;
    asm("msr cpsr, %[ps]" : : [ps] "r"(mode));
    /* set banked stack pointer */
    asm("mov sp, %[ps]" : : [ps] "r"(stack));
}

```

```

    /* go back to SVC mode before executing subroutine return! */
    mode = 0b11010011;
    asm("msr cpsr, %[ps]" : : [ps] "r"(mode));
}

/*
 * Turn on interrupts in the ARM processor
 */
void enable_A9_interrupts(void) {
    int status = 0b01010011;
    asm("msr cpsr, %[ps]" : : [ps] "r"(status));
}

/*
 * Configure the Generic Interrupt Controller (GIC)
 */
void config_GIC(void) {
    config_interrupt (73, 1); // configure the FPGA KEYS interrupt (73)

    // Set Interrupt Priority Mask Register (ICCPMR). Enable interrupts of all
    // priorities
    *((int *) 0xFFFFEC104) = 0xFFFF;

    // Set CPU Interface Control Register (ICCICR). Enable signaling of
    // interrupts
    *((int *) 0xFFFFEC100) = 1;

    // Configure the Distributor Control Register (ICDDCR) to send pending
    // interrupts to CPUs
    *((int *) 0xFFFFED000) = 1;
}

/*
 * Configure Set Enable Registers (ICDISERn) and Interrupt Processor Target
 * Registers (ICDIPTRN). The default (reset) values are used for other registers
 * in the GIC.
 */
void config_interrupt(int N, int CPU_target) {
    int reg_offset, index, value, address;

    /* Configure the Interrupt Set-Enable Registers (ICDISERn).
     * reg_offset = (integer_div(N / 32) * 4
     * value = 1 << (N mod 32) */
    reg_offset = (N >> 3) & 0xFFFFFFF0;
    index      = N & 0x1F;
    value      = 0x1 << index;
    address    = 0xFFFFED100 + reg_offset;
    /* Now that we know the register address and value, set the appropriate bit */
    *((int *)address) |= value;
}

```

```

/* Configure the Interrupt Processor Targets Register (ICDIPTRN)
 * reg_offset = integer_div(N / 4) * 4
 * index = N mod 4 */
reg_offset = (N & 0xFFFFFFFFFC);
index      = N & 0x3;
address    = 0xFFFFED800 + reg_offset + index;
/* Now that we know the register address and value, write to (only) the
 * appropriate byte */
*(char *)address = (char)CPU_target;
}

/*****
 * Pushbutton - Interrupt Service Routine
 *
 * This routine checks which KEY has been pressed. It writes to HEX0
 *****/
void pushbutton_ISR(void) {
    /* KEY base address */
    volatile int * KEY_ptr = (int *) 0xFF200050;
    /* HEX display base address */
    volatile int * HEX3_HEX0_ptr = (int *) 0xFF200020;
    int          press, HEX_bits;

    press        = *(KEY_ptr + 3); // read the pushbutton interrupt register
    *(KEY_ptr + 3) = press;        // Clear the interrupt

    if (press & 0x1)                // KEY0
        HEX_bits = 0b00111111;
    else if (press & 0x2)            // KEY1
        HEX_bits = 0b00000110;
    else if (press & 0x4)            // KEY2
        HEX_bits = 0b01011011;
    else                             // press & 0x8, which is KEY3
        HEX_bits = 0b01001111;

    *HEX3_HEX0_ptr = HEX_bits;
    return;
}

```

Copyright © Intel Corporation. All rights reserved. Intel, the Intel logo, Altera, Arria, Avalon, Cyclone, Enpirion, MAX, Nios, Quartus and Stratix words and logos are trademarks of Intel Corporation or its subsidiaries in the U.S. and/or other countries. Intel warrants performance of its FPGA and semiconductor products to current specifications in accordance with Intel's standard warranty, but reserves the right to make changes to any products and services at any time without notice. Intel assumes no responsibility or liability arising out of the application or use of any information, product, or service described herein except as expressly agreed to in writing by Intel. Intel customers are advised to obtain the latest version of device specifications before relying on any published information and before placing orders for products or services.

*Other names and brands may be claimed as the property of others.